

【声道に対応する共振器 Vocal tract resonance】




Figure 1: 声道に対応する共振器 ブロック図

■差分方程式

入力 $x[n]$, 出力 $y[n]$

$$y[n] = a_1 y[n-1] + a_2 y[n-2] + b_0 x[n]$$

■係数

B_0 : 共振の帯域幅 (振幅特性の値が, 共振周波数における値に対して $\frac{1}{\sqrt{2}}$ になる点間の周波数幅)

F_0 : 共振周波数, T : サンプル間隔

$$a_1 = 2 \exp(-\pi B_0 T) \cos(2\pi F_0 T), \quad a_2 = -\exp(-2\pi B_0 T)$$

$$b_0 = 1 - a_1 - a_2$$

■伝達関数:
$$H(z) = \frac{b_0}{1 - a_1 z^{-1} - a_2 z^{-2}}$$




Figure 2: Scilab 実行結果
 サンプル周波数 8kHz, $B_0 = 50\text{Hz}$, $F_0 = 1, 2, 3\text{kHz}$ 共振器 周波数特性

```

////////////////////////////////////
//      声道共振器
//      Vocal tract resonance
//
//      M. Tsutsui
////////////////////////////////////
clear;

funcprot(0);
function[res]=vocal_tract_res(B,F);//F: 共振周波数 B: 共振の帯域幅

    fs=2*%pi;
    T=1/fs;

    a(1)=2*exp(-%pi*B*T)*cos(2*%pi*F*T);
    a(2)=-exp(-2*%pi*B*T);

    b=1-a(1)-a(2);

    omega=linspace(0,%pi,4000);//周波数 0~4kHzまで描画

    z=exp(%i*omega); //z=exp(jω)

    H=abs(b./(1-a(1)*(z^-1)-a(2)*(z^-2)));//伝達関数 z領域

    res=20*log10(H);

endfunction

F1=%pi/4;//共振周波数 1kHz
F2=%pi/2; //共振周波数 2kHz
F3=3*%pi/4; //共振周波数 3kHz

B=%pi/80;//共振の帯域幅 50Hz

f=linspace(0,4000,4000);//横軸スケーリング(周波数)
plot(f,vocal_tract_res(B,F1));
g=gce();
c=g.children;
c.thickness=3;
plot(f,vocal_tract_res(B,F2),'k');
g=gce();
c=g.children;
c.thickness=3;
plot(f,vocal_tract_res(B,F3),'r');
g=gce();
c=g.children;
c.thickness=3;
xgrid();
legend(['共振周波数_1kHz','共振周波数_2kHz','共振周波数_3kHz']);
title('声道共振器','fontsize',5);

```